// File: contracts/PoolDeployer.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.4;

import "@openzeppelin/contracts/access/Ownable.sol";

import "./interfaces/IMetaPoolFactory.sol";
import "./interfaces/ICryptoPoolFactory.sol";
import "./interfaces/ICryptoPool.sol";
import "./interfaces/IStablePool.sol";
import "./interfaces/IWETH.sol";
import "./interfaces/ITokenClaim.sol";

contract PoolDeployer is Ownable {
    error ZeroAddress();
    error ZeroValue();

    uint256 public constant PRECISION = 1e18;

    uint256 public constant JPEG_PETH_OLD_TOTAL_SUPPLY = 1954993074759827992913957;
    uint256 public constant JPEG_PETH_OLD_VIRTUAL_PRICE = 1005100781903033146;
    uint256 public constant JPEG_PETH_OLD_PRICE_SCALE = 2679005902512888345626761;

    IMetaPoolFactory public immutable METAPOOL_FACTORY;
    ICryptoPoolFactory public immutable CRYPTOPOOL_FACTORY;
    ICryptoPool public immutable JPEG_ETH;
    ICryptoPool public immutable OLD_JPEG_PETH;
    IWETH public immutable WETH;
    IERC20 public immutable PETH_OLD;
    IERC20 public immutable PETH;
    IERC20 public immutable JPEG;

    address public immutable CLAIM_CONTRACT;

    constructor(address _metaPoolFactory, address _cryptoPoolFactory, address _jpegEth, address _oldJpegPeth, address _weth, address _oldPeth, address _newPeth, address _jpeg, address _claimContract) {
        if (
            _metaPoolFactory == address(0) || 
            _cryptoPoolFactory == address(0) || 
            _jpegEth == address(0) || 
            _oldJpegPeth == address(0) ||
            _weth == address(0) || 
            _oldPeth == address(0) || 
            _newPeth == address(0) || 
            _jpeg == address(0) ||
            _claimContract == address(0)
        )
            revert ZeroAddress();

        METAPOOL_FACTORY = IMetaPoolFactory(_metaPoolFactory);
        CRYPTOPOOL_FACTORY = ICryptoPoolFactory(_cryptoPoolFactory);
        JPEG_ETH = ICryptoPool(_jpegEth);
        OLD_JPEG_PETH = ICryptoPool(_oldJpegPeth);
        WETH = IWETH(_weth);
        PETH_OLD = IERC20(_oldPeth);
        PETH = IERC20(_newPeth);
        JPEG = IERC20(_jpeg);

        CLAIM_CONTRACT = _claimContract;
    }
    
    function deployPools(uint256 _oldPethSellAmount, uint256 _targetJPEGPETHSupply) external payable onlyOwner returns (IStablePool _pethWeth, ICryptoPool _jpegPeth) {
        if (msg.value == 0 || _oldPethSellAmount == 0)
            revert ZeroValue();

        PETH_OLD.transferFrom(msg.sender, address(this), _oldPethSellAmount);
        PETH_OLD.approve(address(OLD_JPEG_PETH), _oldPethSellAmount);
        uint256 _receivedLPs = OLD_JPEG_PETH.add_liquidity([0, _oldPethSellAmount], 0, false, address(this));
        IERC20(OLD_JPEG_PETH.token()).approve(address(OLD_JPEG_PETH), _receivedLPs);
        uint256 _receivedJPEG = OLD_JPEG_PETH.remove_liquidity_one_coin(_receivedLPs, 0, 0);

        uint256 _jpegPriceScale = PRECISION * PRECISION / JPEG_ETH.price_scale();
        uint256[2] memory _jpegPETHAmounts = _calculateJPEGPETHBalances(_targetJPEGPETHSupply, _jpegPriceScale);
        
        if (_receivedJPEG > _jpegPETHAmounts[0])
            JPEG.transfer(msg.sender, _receivedJPEG - _jpegPETHAmounts[0]);
        else
            JPEG.transferFrom(msg.sender, address(this), _jpegPETHAmounts[0] - _receivedJPEG);

        PETH.transferFrom(msg.sender, address(this), _jpegPETHAmounts[1] + msg.value);
        WETH.deposit{ value: msg.value }();

        _pethWeth = IStablePool(METAPOOL_FACTORY.deploy_plain_pool(
            "pETH", 
            "pETH-WETH", 
            [address(WETH), address(PETH), address(0), address(0)], 
            3, 
            4000000, 
            1, 
            4
        ));
        WETH.approve(address(_pethWeth), msg.value);
        PETH.approve(address(_pethWeth), msg.value);
        _pethWeth.add_liquidity([msg.value, msg.value], 0, CLAIM_CONTRACT);

        _jpegPeth = ICryptoPool(CRYPTOPOOL_FACTORY.deploy_pool(
            "JPEG/pETH", 
            "JPEGpETH", 
            [address(JPEG), address(PETH)], 
            400000, 
            145000000000000, 
            26000000, 
            45000000, 
            2000000000000, 
            230000000000000, 
            146000000000000, 
            5000000000, 
            600, 
            _jpegPriceScale
        ));
        JPEG.approve(address(_jpegPeth), _jpegPETHAmounts[0]);
        PETH.approve(address(_jpegPeth), _jpegPETHAmounts[1]);
        _jpegPeth.add_liquidity(_jpegPETHAmounts, _targetJPEGPETHSupply, false, CLAIM_CONTRACT);

        ITokenClaim(CLAIM_CONTRACT).enableClaim(address(_pethWeth), _jpegPeth.token());
    }

    function _calculateJPEGPETHBalances(uint256 _newLPTotalSupply, uint256 _newPriceScale) internal pure returns (uint256[2] memory) {
        uint256 _xcp = JPEG_PETH_OLD_TOTAL_SUPPLY * JPEG_PETH_OLD_VIRTUAL_PRICE / PRECISION;
        _xcp = _xcp * _newLPTotalSupply / JPEG_PETH_OLD_TOTAL_SUPPLY;

        uint256 _D = _sqrt(_newPriceScale) * _xcp * 2 / PRECISION;
        return [_D / 2, _D * PRECISION / (_newPriceScale * 2)];
    }

    function _sqrt(uint256 x) internal pure returns (uint256) {
        if (x == 0)
            return 0;
        
        uint256 z = (x + PRECISION) / 2;
        uint256 y = x;

        for (uint256 i; i < 256; ++i) {
            if (z == y)
                return y;
            y = z;
            z = (x * PRECISION / z + z) / 2;
        }

        revert();
    }
}

// File: contracts/interfaces/ITokenClaim.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.4;

interface ITokenClaim {
    function enableClaim(address _pethEthLpToken, address _jpegPethLpToken) external;
}
// File: contracts/interfaces/IWETH.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.4;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface IWETH is IERC20 {

    function deposit() external payable;

}
// File: contracts/interfaces/IStablePool.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.4;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface IStablePool is IERC20 {
    function add_liquidity(uint256[2] calldata _amounts, uint256 _min_mint_amount, address _receiver) external returns (uint256);
}
// File: contracts/interfaces/ICryptoPool.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.4;

interface ICryptoPool {
    function token() external view returns (address);
    function price_scale() external view returns (uint256);
    function add_liquidity(uint256[2] calldata _amounts, uint256 _min_mint_amount, bool _use_eth, address _receiver) external returns (uint256);
    function remove_liquidity_one_coin(uint256 _amount, uint256 _i, uint256 _minAmount) external returns (uint256);
}
// File: contracts/interfaces/ICryptoPoolFactory.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.4;

interface ICryptoPoolFactory {
    function deploy_pool(string calldata _name, string calldata _symbol, address[2] calldata _coins, uint256 _A, uint256 _gamma, uint256 _mid_fee, uint256 _out_fee, uint256 _allowed_extra_profit, uint256 _fee_gamma, uint256 _adjustment_step, uint256 _admin_fee, uint256 _ma_half_time, uint256 _initial_price) external returns (address);
}
// File: contracts/interfaces/IMetaPoolFactory.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.4;

interface IMetaPoolFactory {

    function deploy_plain_pool(string calldata _name, string calldata _symbol, address[4] calldata _coins, uint256 _A, uint256 _fee, uint256 _asset_type, uint256 _implementation_idx) external returns (address);

}
// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}